Linear algebra and Lattices

B. Allombert

IMB
CNRS/Université de Bordeaux

15/06/2022

Vectors and matrices

```
? V = [1,2,3];
? W = [4,5,6]~;
? M = [1,2,3;4,5,6]
%3 =
[1 2 3]
[4 5 6]
? V*W
%17 = 32
? M*W
%18 = [32,77]~
? U = [1..10]
%19 = [1,2,3,4,5,6,7,8,9,10]
```

Components

```
? V[2]
%20 = 2
? W[1..2]
%21 = [4,5]~
? M[2,2]
%22 = 5
? M[1,]
%23 = [1,2,3]
? M[,2]
%24 = [2,5]~
? M[1..2,1..2]
%12 =
[1 2]
[4 5]
```

Constructors

```
? V=vector(10,i,1/i)
%26 = [1,1/2,1/3,1/4,1/5,1/6,1/7,1/8,1/9,1/10]
? W=vectorv(10,i,1/i)
%27 = [1,1/2,1/3,1/4,1/5,1/6,1/7,1/8,1/9,1/10]~
? [1/i | i<-[1..10]]
%28 = [1,1/2,1/3,1/4,1/5,1/6,1/7,1/8,1/9,1/10]
? [1/i | i<-[1..10]]~
%29 = [1,1/2,1/3,1/4,1/5,1/6,1/7,1/8,1/9,1/10]~
? M=matrix(4,4,i,j,i*j)
%30 = [1,2,3,4;2,4,6,8;3,6,9,12;4,8,12,16]
? matdiagonal([1,2,3,4])
%31 = [1,0,0,0;0,2,0,0;0,0,3,0;0,0,0,4]
```

building matrices

Matrices internally are lines of columns.

```
? C1 = [1,2,3]~; C2=[4,5,6]~;
? concat(C1,C2)
%33 = [1,2,3,4,5,6]~
? matconcat([C1,C2])
 [1,4;2,5;3,6]
? matid(5)
? matconcat([C1,C2])
```

linear algebra

Linear algebra follows French convention, matrices act on column vectors on the left.

```
? matdet([1,2;3,4]) \\ determinant
%37 = -2
? M = [1,2,3;4,5,6];
? M~ \\ transposition
%39 = [1,4;2,5;3,6]
? matsize(M) \\ dimensions
%40 = [2,3]
? matrank(M) \\ rank
%41 = 2
? K=matker(M) \\ kernel
%42 = [1;-2;1]
? M*K
%43 = [0;0]
```

LLL reduction

```
? V = vector(10,i,random(10^10));
? M = matconcat([matid(10),V]~);
? T = qflll(M)
%46 = [2,2,3,3,-2,-2,1,5,-1,3;...]
? B = qflll(M,3)
%47 = [2,2,3,3,-2,-2,1,5,-1,3;...]
? M*T==B
%48 = 1
? Q = M~*M;
? U = qflllgram(Q)
%50 = [2,2,3,3,-2,-2,1,5,-1,3;...]
? T == U
%51 = 1
```

Lattices

Example: the Gram matrix of the E8 lattice

```
? E8 = matrix(8,8,i,j,if(i==1&&j==1,4, \
       i==j || (i==1 && j<8) || (j==1 && i<8),2,1))
? E8==E8~ \\ symmetric
%53 = 1
? matdet(E8) \\ unimodular
%54 = 1
? qfsign(E8) \\ signature
%55 = [8,0]
? L = qfminim(E8); L[1..2] \\ 240 minimal vectors o
%56 = [240,2]
? V = L[3][,1] \\ one minimal vector
%57 = [3, -1, -1, -1, -1, -1, -2, 2]~
? qfeval(E8,V) \\ the norm is 2
%58 = 2
```

Lattices

```
? qfperfection(E8) \\ perfection rank
%59 = 36
? G=qfauto(E8); G[1] \\ number of isometries
%60 = 696729600
? A=G[2][1] \\ one isomorphim
%61 = [0,-1,0,-2,0,-1,1,-1;0,1,0,1,0,0,0,1;1,1,0,1,
? A~*E8*A==E8
%62 = 1
```

Lattices and modular forms

```
? [mf,F,C]=mffromqf(E8);
? mfparams(F)
%63 = [1,4,1,y,t-1]
? mfcoefs(F,10)
%64 = [1,240,2160,6720,17520,30240,60480,82560,1404
? mfcoef(F,100003)
%65 = 2400216006480067 20
? L = lfunqf(E8);
? lfunparams(L)
%66 = [1,4,[0,1]]
? lfun(L,0)
?67 = -1;
```

Excercice

Does the same for the Leech lattice:

```
? K=matkerint(Mat(concat([vector(23,i,2*i+1),51,145
? M=matdiagonal(vector(25,i,if(i==25,-1,1)));
? L24 = K~*M*K;
```